
· To ensure that data in the SAP system is protected against unauthorized access and that each user can access only the data for which he or she has explicit authorization, the application developer must implement the SAP R/3 authorization object in each application program.

· For this purpose, you must first develop a logical, application-related authorization model to determine which authorization should be checked during which user action. Then, in the ABAP Workbench, you must create the authorization object and the authorization that suits the model. This object is then assigned to the user with an authorization profile. The application developer must check in the application program – from within the user master of the caller – whether there is a corresponding authorization to control further processing of the program in accordance with the check results. The application developer should do make sure that this authorization check is completed before the action required by the user is performed.

· SAP applied this authorization concept while implementing its software. You, the customer, must also apply this concept when you enhance SAP software or whenever you implement new applications in order to ensure data access authorization. SAP R/3 contains tools that help you to manage authorizations and assign them to user master records.

· An authorization object has a maximum of 10 fields. These fields are not valuated, however.
· An authorization is an authorization object with valuated fields.  You can create several different authorizations for an authorization object. 

· Existing authorizations can be grouped into one authorization profile. An authorization profile should contain all authorizations that are required for executing certain tasks, that is, all the authorizations that are checked in the current user master as to whether they exist when the respective programs or transactions are called. Authorization profiles can be assigned to a user. This contains the authorizations that are contained in the profile.

· You can create authorization objects with required fields and authorizations. You can find the maintenance transaction for authorization fields, authorization objects, and authorizations in the SAP Menu under Tools ( ABAP Workbench ( Development ( Other Tools ( Authorization objects..
· For more information, see the ABAP Editor keyword documentation for authorization concept..

· In the application program, you must perform a check before execution of the required task by the current caller. You must check whether the authorization you have defined as necessary exists so that you can control further processing of the program in accordance with the check result. You can perform this using the AUTHORITY-CHECK statement. You specify an authorization object together with the field valuation. This means that the statement in the user master of the caller checks for the existence of an authorization..

· All the authorization fields and the evaluations must be specified in AUTHORITY-CHECK. If you want to check whether a particular authorization exists and the content of the authorization fields does not matter, write DUMMY instead of FIELD <value> after this authorization field. The check is performed without considering the corresponding field. Here, it is a good idea to perform plausibility checks.

· The AUTHORITY-CHECK supplies return code 0 in the system field sy-subrc after the check only if the caller has the corresponding authorization; otherwise 4 is returned. Based on the return code, you must decide in the program how you want to proceed (read or change data, issue an error message, or some other action). 

· For a full list of all return codes, see the keyword documentation in the ABAP Editor for AUTHORITY-CHECK. 

· In the ABAP Editor, use the template for the AUTHORITY-CHECK statement to include it in your program. In this way, you avoid typing errors that could cause syntax errors or even yield the wrong check results.

· At transaction start, the system automatically checks whether the specified transaction code is known, that is, whether it is marked in the table TSTC and whether this is locked.

· The system then checks whether the caller has the authorization to call the transaction, that is, if he or she has a corresponding authorization for the authorization object S_TCODE.

· Then the system runs through the default authorization checks assigned to the transaction. This means the system checks whether the caller has the default authorization assigned to the transaction.  The assignment Transaction / Default Authorization is specified in the definition of the transaction and stored in the table TSTCA. 

· Only when all these checks have run successfully will the transaction be allowed to start by the system.  Otherwise, processing is terminated with an error message.

· The authorization checks contained in the application program (AUTHORITY-CHECK) are executed only at transaction run time.

· SAP database tables can be buffered at the application server level. The aims of buffering are to:

· Reduce the time needed to access data with read accesses. Data on the application server can be accessed more quickly than data on the database.

· Reduce the load on the database. Reading the data from application server buffers reduces the number of database accesses. 

· The buffered tables are accessed exclusively via database interface mechanisms. 

· When the system executes a Native SQL command, it bypasses the SAP R/3 database interface. Consequently, it does not use the table buffer in SAP R/3 for read or change accesses. You should not, therefore, use Native SQL statements for tables that are buffered, since inconsistencies could occur between the data in the database and that in the buffer.

· Not every Open SQL read access to buffered tables reads buffered data. For information on which operations for read accesses are performed directly on the database, refer to the keyword documentation for SELECT.

· Open SQL commands that change data update the data on the database and the buffers of the application server on which the program updating the database is running. 

· If a table is buffered on several application servers, synchronization of the other buffers is delayed and triggered as follows:

· When data on the database is changed by one of the application servers (more precisely, a program that executes an appropriate OPEN SQL command on one of the application servers) the changed data is registered as such in table DDLOG on the database. The application servers read this table at periodic intervals. If an application server finds relevant entries, the buffer contents are marked in the buffer as being no longer up-to-date.

· The next read access to the data in the buffered table is performed by the database interface on the database. The buffer is updated at the same time. 

· The time between the read accesses to the DDLOG table (invalidation period of the buffers) can be set using the profile parameter, //rdisp/bufreftime.

· There are three types of buffering:

· Resident buffering (100%): The whole table is loaded to the table buffer when the table is accessed for the first time.

· Generic buffering: A generic key (first n key fields) must be specified in the technical settings for the table in the ABAP Dictionary. This key is used to divide the contents of the table into generic areas. If data is accessed using one of the generic keys, the entire generic area is loaded to the table buffer. Client-dependent tables are often buffered generically for each client. 

· Single-record buffering: Only single records are read by the database and stored in the table buffer.

· Native SQL allows you to perform operations on databases that exceed the standard Open SQL command set. Unlike Open SQL, Native SQL supports both operations on the local database, which are active in the SAP system and on decentralized databases.

· Native SQL contains all static statements of the data definition language (DDL) and the data manipulation language (DML) of the relational database system being addressed. Statements for error handling and declaring host variables are not allowed.

· The EXEC SQL and ENDEXEC statements must encapsulate a Native SQL command in an ABAP program. The database table does not have to be declared in the ABAP Dictionary to perform the Native SQL command.

· Since the SQL database language is only partly standardized, you must always use the correct syntax for the Native SQL command from the documentation of the corresponding database manufacturer. 

· For further details on Native SQL, see the keyword documentation in the ABAP Editor under SQL.

· A data cluster is a combination of data objects. Data objects are fields, structured fields, internal tables, and complex structures derived from these. You process data clusters using the ABAP commands EXPORT,IMPORT, and DELETE. Data clusters can be stored in cluster databases.

· You subdivide a cluster database in the ABAP Dictionary into application areas according to your own criteria and using logically related data clusters. The application area name comprises two characters and is freely selectable. You identify the cluster within the application area using an ID (cluster ID).

· For exporting, you require a cluster database. The table INDX is a cluster database for general purposes. Cluster databases should be created as transparent tables in the ABAP Dictionary and must have a standardized structure. For more information, see the online documentation for the EXPORT command.

· For an EXPORT, specify the data objects of your cluster in a list.

· To perform the export, specify the cluster database and the application area within the cluster database. You identify the cluster itself using the cluster ID. If you want to define a data object name in the cluster that is different from the one in the program, use the optional addition FROM. The data objects can be listed in any order. With an export, there is no write protection facility. Existing clusters, therefore, are overwritten if an EXPORT is performed again.

· The data is stored in compressed form in the cluster database.

· You the TABLES statement to declare an application area for your cluster database at the start of your program.

· Important: Header lines in internal tables cannot be exported. Usually, only the table contents are exported.

· If you are working within language constructions from the object-oriented ABAP extension (ABAP Objects), you must use the name substitutions marked as optional on the graphic.

· For an IMPORT, you need to list only a subset of the data objects of your cluster in any order. If you want to define a data object name in the program that differs from the one in the cluster, use the optional addition TO.

· After the IMPORT, the system outputs a return code (sy-subrc). This return code refers to the cluster, rather than to an individual object in the cluster. If the cluster does not exist, the return code is not equal to zero. 

· The structure of the fields, structures, and internal tables to be imported must correspond to the structure of the objects exported to the dataset. If this is not the case, a runtime error occurs. In addition, the objects must be imported using the same name with which they were exported; otherwise, they are not imported. If the cluster exists, the return code is 0, regardless of whether or not objects were imported.

· Important: Here too, only the actual table contents of internal tables are imported; therefore, the header lines remain unchanged. 

· DELETE always deletes the entire cluster. You cannot delete an individual data object within the cluster. After DELETE, the system issues a return code.

· If you are working within language constructions from the object-oriented ABAP extension (ABAP Objects), you must use the name substitutions marked as optional on the graphic.

· You can create your own ABAP cluster databases as follows:

· Define a database table as a transparent table in the ABAP Dictionary. This table represents its cluster database.

· Build the table structure as shown above. 

· The MANDT field can be omitted (it will be filled automatically if it exists).

· The fields RELID, SRTF2, CLUSTR, CLUSTD and the cluster ID are filled automatically within an EXPORT action.

· Have any user-defined fields filled before the EXPORT. They can then be evaluated after an IMPORT.

· Choose the field names for the cluster ID and your own fields. The remaining field names are specified by the system.

· Calculate the length of the part used for the data cluster from the total length of the structure minus the length of the first six fields.

· The INDX database is an example of a database table in which you can store data clusters. It is installed in your system by default and has a key length of 31 bytes.

· You can display the table structure using the keyword help for the table structure INDX. The key consists of a client, area, cluster ID, and subsequent record number. The cluster ID has a default length of 22 bytes, but can have any length in a cluster database.

· With larger data clusters, the runtime system automatically appends subsequent records with the same length. 

· Apart from the key fields and the data cluster, the structure of the database INDX also includes optional fields for administrative information (for example, change, validity date, created by, see structure INDX). You use the SELECT statement to access the key and administration fields, for example, to create a catalog. The system then supplies the administration information fields only if you fill them before the EXPORT using appropriate MOVE statements (for example, MOVE SY-DATUM TO INDX-AEDAT).

· Cluster databases contain the cluster data in the desired form. Large quantities of data can be read via the internal cluster table administration using a limited number of accesses (I/Os).

· Standardization requires that data from several tables be selected in transparent tables, which entails several I/Os. This can, however, be achieved effectively using joins to ABAP Open SQL introduced in release 3.1. The advantage of this is that individual objects can be read from various tables and linked to each other. In cluster tables, however, only the data from one cluster can be read. Links to data in other clusters are not possible.

· Access to cluster data requires knowledge of the cluster ID and the application area. The access also returns, at most, the data of one cluster ID, in other words, it will not return several clusters, as is the case with a SELECT loop. In contrast to this, data can be determined in transparent tables on the basis of any logical expression.

· For _SCOPE = 1, the dialog program contains the locks that it generates. The locks remain set until they are released, either using the function module DEQUEUE_<object>, or implicitly at the end of the program. This includes the ABAP statements LEAVE PROGRAM, LEAVE TO TRANSACTION <ta>, and SUBMIT <program>, and termination messages (message type A).

· If the transaction uses asynchronous update, the update program has no guarantee that the data to be changed is not already locked by another user. For this reason, you should not use _SCOPE = 1 for asynchronous updates.

· If you are using asynchronous update and want to be sure that the locks generated in the dialog program remain set for longer than the V1 update function modules are active, you can use the addition _SCOPE = 3. In this case, the lock is shared between the dialog program and the update program.

· Lock entries that you generate with _SCOPE = 3 must be released both by the dialog program and by the update program.

· Lock entries with _SCOPE = 3 are used only in a few special cases.

· A Business Application Programming Interface (BAPI) functions as a gateway to SAP R/3, providing access to its business data and processes. 

· An object in the BOR can have many methods, one or more of which are implemented as BAPIs.

· BAPIs can have various functions, including:

· Generating an object

· Querying attributes of an object

· Changing attributes of an object

· A BAPI is an interface that can be used for various applications, for example:

· Internet Application Components – Mapping individual SAP R/3 functions on the Internet or intranet for users who have no SAP R/3 experience

· SAP R/3 component formation - Communication between the business objects of different SAP R/3 components (applications)

· VisualBasic/JAVA/C++ - External clients (for example, alternative GUIs) access business data and processes directly

For BAPIs with UPDATE semantics:

· BAPIs are always called up synchronously.
Exception: BAPIs are sent asynchronously via ALE in an IDoc and then called up again synchronously in the target system (new as of Release 4.0)

· BAPIs implement your database changes through the update task.

· New as of Release 4.6: BAPIs themselves do not program a COMMIT WORK. Instead, you must use the following methods of the service object BAPISERVICE (technical name SAP0001):
- TransactionCommit (FBS name: BAPI_TRANSACTION_COMMIT) 
- TransactionRollback (FBS name: BAPI_TRANSACTION_ROLLBACK) 

· In release 3.1, the BAPIs themselves carry out the COMMIT WORK command; in other words, a BAPI is synonymous with an LUW or transaction. 

· If a BAPI itself carries out a COMMIT WORK command, it must be listed explicitly in the documentation for the BAPI (as of Release 4.0). This is the only way the user can find out that a COMMIT WORK takes place in the BAPI.

· These BAPIs must also be documented in the Online Service System in note no. 0131838, collective note for BAPIs with Commit Work commands.

· In release 4.0A, the Commit control must be removed from the writing BAPIs, in other words, from the BAPIs that cause the database changes.  Here, the existing transaction model used in release 3.1 should not be changed. For this purpose, the RFC-capable function module BAPI_TRANSACTION_COMMIT, which performs the COMMIT WORK command, must be called.

· In the transaction model used to develop BAPIs, a transaction represents a processing unit or logical unit of work (LUW).

· Operations that change the database must be carried out by the update task.

· As of release 4.0, BAPIs no longer carry out COMMIT WORK commands. This enables several BAPIs to be combined with each other in an LUW. 

· To close LUW processing initiated by the BAPI calls, use the function module BAPI_TRANSACTION_COMMIT. This funcion module implements the TransactionCommit method of the BAPISERVICE business object.

· For further details on using BAPIs, see the BAPI User Manual and the BAPI Programming Guide. These are available in the online documentation and on the SAP Homepage (http://www.sap-ag.de) under Technology ( Open BAPI Network/ BAPI Section.


[image: image34.wmf] 

ã

 

SAP AG 2003

 

Overview of Additional Slides 7

 

SAP Buffers

 

SAP Buffers

 

Native SQL

 

Native SQL

 

Cluster Tables

 

Cluster Tables

 

SAP Locks

 

SAP Locks

 

BAPI Transaction Model

 

BAPI Transaction Model

 

COMMIT

 

WORK and

 ROLLBACK WORK

 

COMMIT

 

WORK and ROLLBACK WORK

 

Authorization Checks

 

Authorization Checks

 


[image: image35.wmf]ã

SAP

AG 2002

Actions for

COMMIT WORK

l

Actions on the database

n

All

updates that were made during the current dialog step are commi

tted

n

All

database locks are released

n

All

open database cursors are closed

(

even those that were opened using

WITH HOLD) 

l

All

subroutines registered using

PERFORM ON COMMIT

are executed

l

Update

is triggered

(

CALL FUNCTION IN UPDATE TASK

)

n

COMMIT WORK

-

does

not

wait until the

end

of the update

n

COMMIT WORK

-

waits until the

end

of the update

n

For

a

local update

:

Immediate execution of the update modules

l

Locks

( _SCOPE = 2 )

are

:

n

NOT

released if no update requests have come

up

n

Transferred

to

the update and released when it has ended

(

as

a

rule

)


[image: image36.wmf]ã

SAP

AG 2002

l

Current

DB LUW (DB Rollback)

is ended

n

All

changes made

in

the 

current

dialog step are undone

n

All

database locks are released

n

All

open cursors are closed

l

Data

in

the corresponding

SAP LUW

is deleted

:

n

Registration of update modules registered with

CALL FUNCTION 

IN UPDATE TASK

is rejected

n

The same applies

to

the subroutines registered with

PERFORM 

ON COMMIT

n

Function modules that were registered for transactional or 

queued

RFC (CALL FUNCTION IN BACKGROUND TASK)

are

not

executed

l

SAP

locks are released

( _

scope

= 2 )

Actions for

ROLLBACK WORK


[image: image37.wmf]ã

SAP

AG 2002

l

Undo updates that were committed previously

l

Undo updates

to

internal tables and other data objects

(

program context

)

l

Reset values

in

calculated 'contexts'

l

Undo changes made

to

operating system files

l

Undo actions that were executed during synchronous RFCs

Actions That

a ROLLBACK WORK

Does 

Not

Perform


[image: image38.wmf]ã

SAP

AG 2002

SAP LUW:

Overall

View

R

Legend

A

forces

B

on basis of

R

Performance

requirements of the system

Database

performance

Appl

.

server

performance

Transaction

concept

Appl

.

server

architecture

DB LUW <=

Dialog

step

Changes

to

a DB LUW

Lock

mechanism

Functional requirements

SAP LUW

concept

Fewer database

users as

a

result of

dispatching

One work process

for several users

"

at the

same time

"

DB LUW

independent of

user

dialogs

Fewer

DB

users

"

writing

"

as

a

result of async

.

update

Optimal

use of

resources

(CPU,

storage

, ...)

Everything

will

apply

"

at the same time

"

Mandatory

Mandatory


Solutions

	[image: image39.png]



	Program: 
Generating Customer Data Records




Complete Transaction


Model Solution SAPBC414S_CREATE_CUSTOMER

1 Module Pool

*&-----------------------------------------------------------------*

*& Modulpool         SAPBC414S_CREATE_CUSTOMER                     *
*&-----------------------------------------------------------------*

INCLUDE BC414S_CREATE_CUSTOMERTOP.
INCLUDE BC414S_CREATE_CUSTOMERO01.
INCLUDE BC414S_CREATE_CUSTOMERI01.
INCLUDE BC414S_CREATE_CUSTOMERF01.
1.2 SCREEN 100
PROCESS BEFORE OUTPUT.
  MODULE status_0100.
PROCESS AFTER INPUT.
  MODULE exit AT EXIT-COMMAND.
  MODULE save_ok_code.
  FIELD: scustom-name MODULE mark_changed ON REQUEST.
  MODULE user_command_0100.
1.3 TOP Include
*&-----------------------------------------------------------------*

*& Include BC414S_CREATE_CUSTOMERTOP                               *
*&-----------------------------------------------------------------*

PROGRAM  sapbc414s_create_customer MESSAGE-ID bc414.
DATA: answer, flag.
DATA: ok_code LIKE sy-ucomm, save_ok LIKE ok_code.
TABLES: scustom.
PBO Module
*------------------------------------------------------------------*

***INCLUDE BC414S_CREATE_CUSTOMERO01 .
*------------------------------------------------------------------*

*&-----------------------------------------------------------------*

*&      Module  STATUS_0100  OUTPUT
*&-----------------------------------------------------------------*

MODULE STATUS_0100 OUTPUT.
  SET PF-STATUS 'DYN_0100'.
  SET TITLEBAR 'DYN_0100'.
ENDMODULE.                 " STATUS_0100  OUTPUT
1.4 PAI Module
*------------------------------------------------------------------*

***INCLUDE BC414S_CREATE_CUSTOMERI01 .
*------------------------------------------------------------------*

*&-----------------------------------------------------------------*

*&      Module  EXIT  INPUT
*&-----------------------------------------------------------------*

MODULE exit INPUT.
  CASE ok_code.
    WHEN 'EXIT'.
      IF sy-datar IS INITIAL AND flag IS INITIAL.
* no changes on screen 100
        LEAVE PROGRAM.
      ELSE.
        PERFORM ask_save USING answer.
        CASE answer.
          WHEN 'J'.
            ok_code = 'SAVE&EXIT'.
          WHEN 'N'.
            LEAVE PROGRAM.
          WHEN 'A'.
            CLEAR ok_code.
            SET SCREEN 100.
        ENDCASE.
      ENDIF.
    WHEN 'CANCEL'.
      IF sy-datar IS INITIAL AND flag IS INITIAL.
* no changes on screen 100
        LEAVE TO SCREEN 0.
      ELSE.
        PERFORM ask_loss USING answer.
        CASE answer.
          WHEN 'J'.
            LEAVE TO SCREEN 0.
          WHEN 'N'.
            CLEAR ok_code.
            SET SCREEN 100.
        ENDCASE.
      ENDIF.
  ENDCASE.
ENDMODULE.                             " EXIT  INPUT
*&-----------------------------------------------------------------*

*&      Module  SAVE_OK_CODE  INPUT
*&-----------------------------------------------------------------*

MODULE save_ok_code INPUT.
  save_ok = ok_code.
  CLEAR ok_code.
ENDMODULE.                             " SAVE_OK_CODE  INPUT
*&-----------------------------------------------------------------*

*&      Module  USER_COMMAND_0100  INPUT
*&-----------------------------------------------------------------*

MODULE user_command_0100 INPUT.
  CASE save_ok.
    WHEN 'SAVE&EXIT'.
      PERFORM save.
      LEAVE PROGRAM.
    WHEN 'SAVE'.
      IF flag IS INITIAL.
        SET SCREEN 100.
      ELSE.
        PERFORM save.
        SET SCREEN 0.
      ENDIF.
    WHEN 'BACK'.
      IF flag IS INITIAL.
        SET SCREEN 0.
      ELSE.
        PERFORM ask_save USING answer.
        CASE answer.
          WHEN 'J'.
            PERFORM save.
            SET SCREEN 0.
          WHEN 'N'.
            SET SCREEN 0.
          WHEN 'A'.
            SET SCREEN 100.
        ENDCASE.
      ENDIF.
  ENDCASE.
ENDMODULE.                             " USER_COMMAND_0100  INPUT
*&-----------------------------------------------------------------*

*&      Module  MARK_CHANGED  INPUT
*&-----------------------------------------------------------------*

MODULE mark_changed INPUT.
* set flag to mark changes were made on screen 100
  flag = 'X'.
ENDMODULE.                             " MARK_CHANGED  INPUT
FORM Routines

*------------------------------------------------------------------*

***INCLUDE BC414S_CREATE_CUSTOMERF01 .
*------------------------------------------------------------------*

*&-----------------------------------------------------------------*

*&      Form  NUMBER_GET_NEXT
*&-----------------------------------------------------------------*

*      -->P_WA_SCUSTOM  text
*------------------------------------------------------------------*

FORM number_get_next USING p_scustom LIKE scustom.
  DATA: return TYPE inri-returncode.
* get next free number in the number range '01'
* of number range object 'SBUSPID'
  CALL FUNCTION 'NUMBER_GET_NEXT'
       EXPORTING
            nr_range_nr = '01'
            object      = 'SBUSPID'
       IMPORTING
            number      = p_scustom-id
            returncode  = return
       EXCEPTIONS
            OTHERS      = 1.
  CASE sy-subrc.
    WHEN 0.
      CASE return.
        WHEN 1.
* number of remaining numbers critical
          MESSAGE s070.
        WHEN 2.
* last number
          MESSAGE s071.
        WHEN 3.
* no free number left over
          MESSAGE a072.
      ENDCASE.
    WHEN 1.
* internal error
      MESSAGE a073 WITH sy-subrc.
  ENDCASE.
ENDFORM.                               " NUMBER_GET_NEXT
*&-----------------------------------------------------------------*

*&      Form  ASK_SAVE
*&-----------------------------------------------------------------*

*      -->P_ANSWER  text
*------------------------------------------------------------------*

FORM ask_save USING p_answer.
  CALL FUNCTION 'POPUP_TO_CONFIRM_STEP'
       EXPORTING
            textline1 = 'Data has been changed.'(001)
            textline2 = 'Save before leaving transaction?'(002)
            titel     = 'Create Customer'(003)
       IMPORTING
            answer    = p_answer.
ENDFORM.                               " ASK_SAVE
*&-----------------------------------------------------------------*

*&      Form  ASK_LOSS
*&-----------------------------------------------------------------*

*      -->P_ANSWER  text
*------------------------------------------------------------------*

FORM ask_loss USING p_answer.
  CALL FUNCTION 'POPUP_TO_CONFIRM_LOSS_OF_DATA'
       EXPORTING
            textline1 = 'Continue?'(004)
            titel     = 'Create Customer'(003)
       IMPORTING
            answer    = p_answer.
ENDFORM.                               " ASK_LOSS
*&-----------------------------------------------------------------*

*&      Form  ENQ_SCUSTOM
*&-----------------------------------------------------------------*

FORM enq_scustom.
  CALL FUNCTION 'ENQUEUE_ESCUSTOM'
       EXPORTING
            id             = scustom-id
       EXCEPTIONS
            foreign_lock   = 1
            system_failure = 2
            OTHERS         = 3.
  CASE sy-subrc.
    WHEN 1.
* dataset allready locked
      MESSAGE e060.
    WHEN 2 OR 3.
* locking of dataset not possible for other reasons
      MESSAGE e063 WITH sy-subrc.
  ENDCASE.
ENDFORM.                               " ENQ_SCUSTOM
*&-----------------------------------------------------------------*

*&      Form  DEQ_ALL
*&-----------------------------------------------------------------*

FORM deq_all.
  CALL FUNCTION 'DEQUEUE_ALL'.
ENDFORM.                               " DEQ_ALL
*&-----------------------------------------------------------------*

*&      Form  SAVE_SCUSTOM
*&-----------------------------------------------------------------*

FORM save_scustom.
  INSERT INTO scustom VALUES scustom.
  IF sy-subrc <> 0.
    MESSAGE a048.
  ELSE.
    SET PARAMETER ID 'CSM' FIELD scustom-id.
    MESSAGE s015 WITH scustom-id.
  ENDIF.
ENDFORM.                               " SAVE_SCUSTOM
*&-----------------------------------------------------------------*

*&      Form  SAVE
*&-----------------------------------------------------------------*

FORM save.
* get SCUSTOM-ID from number range object SBUSPID
  PERFORM number_get_next USING scustom.
* lock dataset
  PERFORM enq_scustom.
* save new customer
  PERFORM save_scustom.
* unlock dataset
  PERFORM deq_all.
ENDFORM.                               " SAVE
Solutions

	Program:

Canceling/Creating Bookings




Complete Transaction


Model Solution SAPBC414S_BOOKINGS

2 Module Pool

*&-----------------------------------------------------------------*

*& Modulpool         SAPBC414S_BOOKINGS                            *
*&-----------------------------------------------------------------*

INCLUDE bc414s_bookingstop.
INCLUDE bc414s_bookingso01.
INCLUDE bc414s_bookingsi01.
INCLUDE bc414s_bookingsf01.
INCLUDE bc414s_bookingsf02.
INCLUDE bc414s_bookingsf03.
INCLUDE bc414s_bookingsf04.
INCLUDE bc414s_bookingsf05.
INCLUDE bc414s_bookingsf06.
INCLUDE fbc414_cdocscdc.
3 SCREEN 100
PROCESS BEFORE OUTPUT.
  MODULE STATUS_0100.
*

PROCESS AFTER INPUT.
  MODULE EXIT AT EXIT-COMMAND.
  MODULE SAVE_OK_CODE.
  CHAIN.
* cancel booking: check if flight exists or flight can be created
    FIELD: SDYN_CONN-CARRID, SDYN_CONN-CONNID, SDYN_CONN-FLDATE.
    MODULE USER_COMMAND_0100.
  ENDCHAIN.
4 SCREEN 200
PROCESS BEFORE OUTPUT.
  MODULE STATUS_0200.
  MODULE TRANS_DETAILS.
  CALL SUBSCREEN SUB1 INCLUDING SY-CPROG '0201'.
  LOOP AT ITAB_BOOK INTO WA_BOOK WITH CONTROL TC_SBOOK.
    MODULE TRANS_TO_TC.
* allow only modification of bookings, that are not allready 
* cancelled
    MODULE MODIFY_SCREEN.
  ENDLOOP.
*

PROCESS AFTER INPUT.
  LOOP AT ITAB_BOOK.
* mark changed bookings in internal table itab_book
    FIELD SDYN_BOOK-CANCELLED MODULE MODIFY_ITAB ON REQUEST.
  ENDLOOP.
  MODULE EXIT AT EXIT-COMMAND.
  MODULE SAVE_OK_CODE.
  MODULE USER_COMMAND_0200.
5 SCREEN 201
PROCESS BEFORE OUTPUT.
PROCESS AFTER INPUT.
6 SCREEN 300
PROCESS BEFORE OUTPUT.
  MODULE STATUS_0300.
  MODULE TABSTRIP_INIT.
  MODULE TRANS_DETAILS.
  CALL SUBSCREEN TAB_SUB INCLUDING SY-CPROG SCREEN_NO.
*

PROCESS AFTER INPUT.
  CALL SUBSCREEN TAB_SUB.
  MODULE EXIT AT EXIT-COMMAND.
  MODULE SAVE_OK_CODE.
  MODULE TRANS_FROM_0300.
  MODULE USER_COMMAND_0300.
7 SCREEN 301
PROCESS BEFORE OUTPUT.
  MODULE HIDE_BOOKID.
PROCESS AFTER INPUT.
8 SCREEN 302
PROCESS BEFORE OUTPUT.
PROCESS AFTER INPUT.
9 SCREEN 303
PROCESS BEFORE OUTPUT.
PROCESS AFTER INPUT.
10 TOP Include
*&-----------------------------------------------------------------*

*& Include BC414S_BOOKINGSTOP                                      *
*&-----------------------------------------------------------------*

PROGRAM  sapbc414s_bookings MESSAGE-ID bc414.
* change documents: data definitions for use of function modules
INCLUDE fbc414_cdocscdt.
* line type of internal table itab_book, used to display bookings in
* table control
TYPES: BEGIN OF wa_book_type.
         INCLUDE STRUCTURE sbook.
TYPES:   name TYPE scustom-name,
         mark,
       END OF wa_book_type.
* work area and internal table used to display bookings in table 
* control
DATA: wa_book TYPE wa_book_type,
      itab_book TYPE TABLE OF wa_book_type.
* bookings to be modified on database table
DATA: itab_sbook_modify TYPE TABLE OF sbook.
* change documents: bookings before changes are performed
DATA: itab_cd TYPE TABLE OF sbook WITH NON-UNIQUE KEY
      carrid connid fldate bookid customid.
* work areas for database tables spfli, sflight, sbook.
DATA: wa_sbook TYPE sbook,
      wa_sflight TYPE sflight,
      wa_spfli TYPE spfli.
* complex transaction: customer ID created in the called transaction
data: scust_id(20).
* transport function codes from screens
DATA: ok_code TYPE sy-ucomm, save_ok LIKE ok_code.
* define subscreen screen number on tabstrip, screen 300
DATA: screen_no TYPE sy-dynnr.
* used to handle sy-subrc, which is determined in subroutine
DATA  sysubrc LIKE sy-subrc.
* For field transport to/from screen
TABLES: sdyn_conn, sdyn_book.
* table control declaration (display bookings),
* tabstrip declaration (create booking)
CONTROLS: tc_sbook TYPE TABLEVIEW USING SCREEN '0200',
          tab TYPE TABSTRIP.
11 PBO Module
*------------------------------------------------------------------*

***INCLUDE BC414S_BOOKINGSO01 .
*------------------------------------------------------------------*

*&-----------------------------------------------------------------*

*&      Module  STATUS_0100  OUTPUT
*&-----------------------------------------------------------------*

MODULE status_0100 OUTPUT.
  SET PF-STATUS 'DYN_100'.
  SET TITLEBAR 'DYN_100'.
ENDMODULE.                             " STATUS_0100  OUTPUT
*&-----------------------------------------------------------------*

*&      Module  STATUS_0200  OUTPUT
*&-----------------------------------------------------------------*

MODULE status_0200 OUTPUT.
  SET PF-STATUS 'DYN_200'.
  SET TITLEBAR 'DYN_200' WITH sdyn_conn-carrid
                              sdyn_conn-connid
                              sdyn_conn-fldate.
ENDMODULE.                             " STATUS_0200  OUTPUT
*&-----------------------------------------------------------------*

*&      Module  STATUS_0300  OUTPUT
*&-----------------------------------------------------------------*

MODULE status_0300 OUTPUT.
  SET PF-STATUS 'DYN_300'.
  SET TITLEBAR 'DYN_300' WITH sdyn_conn-carrid
                              sdyn_conn-connid
                              sdyn_conn-fldate.
ENDMODULE.                             " STATUS_0300  OUTPUT
*&-----------------------------------------------------------------*

*&      Module  TRANS_DETAILS  OUTPUT
*&-----------------------------------------------------------------*

MODULE trans_details OUTPUT.
  MOVE-CORRESPONDING: wa_spfli   TO sdyn_conn,
                      wa_sflight TO sdyn_conn,
                      wa_sbook   TO sdyn_book.
ENDMODULE.                             " TRANS_DETAILS  OUTPUT
*&-----------------------------------------------------------------*

*&      Module  TRANS_TO_TC  OUTPUT
*&-----------------------------------------------------------------*

MODULE trans_to_tc OUTPUT.
  MOVE-CORRESPONDING wa_book TO sdyn_book.
ENDMODULE.                             " TRANS_TO_TC  OUTPUT
*&-----------------------------------------------------------------*

*&      Module  MODIFY_SCREEN  OUTPUT
*&-----------------------------------------------------------------*

MODULE modify_screen OUTPUT.
  LOOP AT SCREEN.
    CHECK screen-name = 'SDYN_BOOK-CANCELLED'.
    CHECK ( NOT sdyn_book-cancelled IS INITIAL ) AND
          ( sdyn_book-mark IS INITIAL ).
    screen-input = 0.
    MODIFY SCREEN.
  ENDLOOP.
ENDMODULE.                             " MODIFY_SCREEN  OUTPUT
*&-----------------------------------------------------------------*

*&      Module  TABSTRIP_INIT  OUTPUT
*&-----------------------------------------------------------------*

MODULE tabstrip_init OUTPUT.
  CHECK tab-activetab IS INITIAL.
  tab-activetab = 'BOOK'.
  screen_no = '0301'.
ENDMODULE.                             " TABSTRIP_INIT  OUTPUT
*&-----------------------------------------------------------------*

*&      Module  HIDE_BOOKID  OUTPUT
*&-----------------------------------------------------------------*

MODULE hide_bookid OUTPUT.
* hide field displaying customer number when working with number range
* object BS_SCUSTOM
  LOOP AT SCREEN.
    CHECK screen-name = 'SDYN_BOOK-BOOKID'.
    screen-active = 0.
    MODIFY SCREEN.
  ENDLOOP.
ENDMODULE.                             " HIDE_BOOKID  OUTPUT
12 PAI Module
*------------------------------------------------------------------*

***INCLUDE BC414S_BOOKINGSI01 .
*------------------------------------------------------------------*

*&-----------------------------------------------------------------*

*&      Module  EXIT  INPUT
*&-----------------------------------------------------------------*

MODULE exit INPUT.
  CASE ok_code.
    WHEN 'CANCEL'.
      CASE sy-dynnr.
        WHEN '0100'.
          LEAVE PROGRAM.
        WHEN '0200'.
          PERFORM deq_all.
          LEAVE TO SCREEN '0100'.
        WHEN '0300'.
          PERFORM deq_all.
          LEAVE TO SCREEN '0100'.
        WHEN OTHERS.
      ENDCASE.
    WHEN 'EXIT'.
      LEAVE PROGRAM.
    WHEN OTHERS.
  ENDCASE.
ENDMODULE.                             " EXIT  INPUT
*&-----------------------------------------------------------------*

*&      Module  SAVE_OK_CODE  INPUT
*&-----------------------------------------------------------------*

MODULE save_ok_code INPUT.
  save_ok = ok_code.
  CLEAR ok_code.
ENDMODULE.                             " SAVE_OK_CODE  INPUT
*&-----------------------------------------------------------------*

*&      Module  USER_COMMAND_0100  INPUT
*&-----------------------------------------------------------------*

MODULE user_command_0100 INPUT.
  CASE save_ok.
****************************CANCEL BOOKING**************************
    WHEN 'BOOKC'.
      PERFORM enq_sflight_sbook.
      PERFORM read_sflight USING wa_sflight sysubrc.
* process returncode - if flight does not exist: e-message
      PERFORM process_sysubrc_bookc.
      PERFORM read_spfli USING wa_spfli.
      PERFORM read_sbook USING itab_book itab_cd.
      REFRESH CONTROL 'TC_SBOOK' FROM SCREEN '0200'.
****************************CREATE BOOKING**************************
    WHEN 'BOOKN'.
      PERFORM enq_sflight.
      PERFORM read_sflight USING wa_sflight sysubrc.
* process returncode - if flight does not exist: e-message
      PERFORM process_sysubrc_bookn.
      PERFORM read_spfli USING wa_spfli.
      PERFORM initialize_sbook USING wa_sbook.
    WHEN 'BACK'.
      SET SCREEN 0.
    WHEN OTHERS.
      SET SCREEN '0100'.
  ENDCASE.
ENDMODULE.                             " USER_COMMAND_0100  INPUT
*&-----------------------------------------------------------------*

*&      Module  USER_COMMAND_0200  INPUT
*&-----------------------------------------------------------------*

MODULE user_command_0200 INPUT.
  CASE save_ok.
    WHEN 'SAVE'.
* collect marked (changed) data sets in seperate internal table
      PERFORM collect_modified_data USING itab_sbook_modify.
* perform database changes
      PERFORM save_modified_booking.
* create change documents
      PERFORM create_change_documents.
      COMMIT WORK.
* Unlocking data sets is executed by the update program !!
      SET SCREEN '0100'.
    WHEN 'BACK'.
      PERFORM deq_all.
      SET SCREEN '0100'.
    WHEN OTHERS.
      SET SCREEN '0200'.
  ENDCASE.
ENDMODULE.                             " USER_COMMAND_0200  INPUT
*&-----------------------------------------------------------------*

*&      Module  MODIFY_ITAB  INPUT
*&-----------------------------------------------------------------*

MODULE modify_itab INPUT.
  wa_book-cancelled = sdyn_book-cancelled.
  wa_book-mark = 'X'.
  MODIFY itab_book FROM wa_book INDEX tc_sbook-current_line.
ENDMODULE.                             " MODIFY_ITAB  INPUT
*&-----------------------------------------------------------------*

*&      Module  USER_COMMAND_0300  INPUT
*&-----------------------------------------------------------------*

MODULE user_command_0300 INPUT.
  PERFORM tabstrip_set.
  CASE save_ok.
    WHEN 'NEW_CUSTOM'.
      PERFORM create_new_customer.
      SET SCREEN '0300'.
    WHEN 'SAVE'.
      PERFORM save_new_booking.
      COMMIT WORK.
* Unlocking data sets is executed by the update program !!
      SET SCREEN '0100'.
    WHEN 'BACK'.
      PERFORM deq_all.
      SET SCREEN '0100'.
    WHEN OTHERS.
      SET SCREEN '0300'.
  ENDCASE.
ENDMODULE.                             " USER_COMMAND_0300  INPUT
*&-----------------------------------------------------------------*

*&      Module  TRANS_FROM_0300  INPUT
*&-----------------------------------------------------------------*

MODULE trans_from_0300 INPUT.
  MOVE-CORRESPONDING sdyn_book TO wa_sbook.
ENDMODULE.                             " TRANS_FROM_0300  INPUT
13 FORM Routines

14 F01
*------------------------------------------------------------------*

***INCLUDE BC414S_BOOKINGSF01 .
*------------------------------------------------------------------*

*&-----------------------------------------------------------------*

*&      Form  COLLECT_MODIFIED_DATA
*&-----------------------------------------------------------------*

*      -->P_ITAB_SBOOK_MODIFY  text
*------------------------------------------------------------------*

FORM collect_modified_data USING p_itab_sbook_modify
                                 LIKE itab_sbook_modify.
  DATA: wa_book LIKE LINE OF itab_book,
        wa_sbook_modify LIKE LINE OF p_itab_sbook_modify.
  CLEAR: p_itab_sbook_modify.
* Only bookings are collected, that
* 1) have been changed (mark = 'X')
* 2) shall be cancelled (cancelled = 'X')
  LOOP AT itab_book INTO wa_book
       WHERE    mark = 'X'
       AND cancelled = 'X'.
    MOVE-CORRESPONDING wa_book TO wa_sbook_modify.
    APPEND wa_sbook_modify TO p_itab_sbook_modify.
  ENDLOOP.
ENDFORM.                               " COLLECT_MODIFIED_DATA
*&-----------------------------------------------------------------*

*&      Form  INITIALIZE_SBOOK
*&-----------------------------------------------------------------*

*      -->P_WA_SBOOK  text
*------------------------------------------------------------------*

FORM initialize_sbook USING p_wa_sbook TYPE sbook.
  CLEAR p_wa_sbook.
  MOVE-CORRESPONDING wa_sflight TO p_wa_sbook.
  MOVE: wa_sflight-price    TO p_wa_sbook-forcurram,
        wa_sflight-currency TO p_wa_sbook-forcurkey,
        sy-datum            TO p_wa_sbook-order_date.
ENDFORM.                               " INITIALIZE_SBOOK
*&-----------------------------------------------------------------*

*&      Form  PROCESS_SYSUBRC_BOOKC
*&-----------------------------------------------------------------*

FORM process_sysubrc_bookc.
  CASE sysubrc.
    WHEN 0.
      SET SCREEN '0200'.
    WHEN OTHERS.
      PERFORM deq_all.
      MESSAGE e023 WITH sdyn_conn-carrid sdyn_conn-connid
                        sdyn_conn-fldate.
  ENDCASE.
ENDFORM.                               " PROCESS_SYSUBRC_BOOKC
*&-----------------------------------------------------------------*

*&      Form  PROCESS_SYSUBRC_BOOKN
*&-----------------------------------------------------------------*

FORM process_sysubrc_bookn.
  CASE sysubrc.
    WHEN 0.
      SET SCREEN '0300'.
    WHEN OTHERS.
      PERFORM deq_all.
      MESSAGE e023 WITH sdyn_conn-carrid sdyn_conn-connid
                        sdyn_conn-fldate.
  ENDCASE.
ENDFORM.                               " PROCESS_SYSUBRC_BOOKN
*&-----------------------------------------------------------------*

*&      Form  TABSTRIP_SET
*&-----------------------------------------------------------------*

FORM tabstrip_set.
  IF save_ok = 'BOOK' OR save_ok = 'DETCON' OR save_ok = 'DETFLT'.
    tab-activetab = save_ok.
  ENDIF.
  CASE save_ok.
    WHEN 'BOOK'.
      screen_no = '0301'.
    WHEN 'DETCON'.
      screen_no = '0302'.
    WHEN 'DETFLT'.
      screen_no = '0303'.
  ENDCASE.
ENDFORM.                               " TABSTRIP_SET
*&-----------------------------------------------------------------*

*&      Form  NUMBER_GET_NEXT
*&-----------------------------------------------------------------*

*      -->P_WA_SBOOK  text
*------------------------------------------------------------------*

FORM number_get_next USING p_wa_sbook LIKE sbook.
  DATA: return TYPE inri-returncode.
* get next free number in the number range '01' of number range 
* object 'SBOOKID'
  CALL FUNCTION 'NUMBER_GET_NEXT'
       EXPORTING
            nr_range_nr = '01'
            object      = 'SBOOKID'
            subobject   = p_wa_sbook-carrid
       IMPORTING
            number      = p_wa_sbook-bookid
            returncode  = return
       EXCEPTIONS
            OTHERS      = 1.
  CASE sy-subrc.
    WHEN 0.
      CASE return.
        WHEN 1.
* number of remaining numbers critical
          MESSAGE s070.
        WHEN 2.
* last number
          MESSAGE s071.
        WHEN 3.
* no free number left over
          MESSAGE a072.
      ENDCASE.
    WHEN 1.
* internal error
      MESSAGE a073 WITH sy-subrc.
  ENDCASE.
ENDFORM.                               " NUMBER_GET_NEXT
*&-----------------------------------------------------------------*

*&      Form  CREATE_NEW_CUSTOMER
*&-----------------------------------------------------------------*

FORM create_new_customer.
  CALL TRANSACTION 'BC414S_CREATE_CUST'.
  GET PARAMETER ID 'CSM' FIELD wa_sbook-customid.
* Alternative solution using the GET function of the screen field
* for customer ID
*   CLEAR wa_sbook-customid.
ENDFORM.                               " CREATE_NEW_CUSTOMER
15 F02
*------------------------------------------------------------------*

*   INCLUDE BC414S_BOOKINGSF02
*------------------------------------------------------------------*

*&-----------------------------------------------------------------*

*&      Form  ENQ_SFLIGHT
*&-----------------------------------------------------------------*

FORM enq_sflight.
  CALL FUNCTION 'ENQUEUE_ESFLIGHT'
       EXPORTING
            carrid         = sdyn_conn-carrid
            connid         = sdyn_conn-connid
            fldate         = sdyn_conn-fldate
       EXCEPTIONS
            foreign_lock   = 1
            system_failure = 2
            OTHERS         = 3.
  CASE sy-subrc.
    WHEN 0.
    WHEN 1.
      MESSAGE e060.
    WHEN OTHERS.
      MESSAGE e063 WITH sy-subrc.
  ENDCASE.
ENDFORM.                               " ENQ_SFLIGHT
*&-----------------------------------------------------------------*

*&      Form  ENQ_SFLIGHT_SBOOK
*&-----------------------------------------------------------------*

FORM enq_sflight_sbook.
  CALL FUNCTION 'ENQUEUE_ESFLIGHT_SBOOK'
       EXPORTING
            carrid         = sdyn_conn-carrid
            connid         = sdyn_conn-connid
            fldate         = sdyn_conn-fldate
       EXCEPTIONS
            foreign_lock   = 1
            system_failure = 2
            OTHERS         = 3.
  CASE sy-subrc.
    WHEN 0.
    WHEN 1.
      MESSAGE e062.
    WHEN OTHERS.
      MESSAGE e063 WITH sy-subrc.
  ENDCASE.
ENDFORM.                               " ENQ_SFLIGHT_SBOOK
*&-----------------------------------------------------------------*

*&      Form  ENQ_SBOOK
*&-----------------------------------------------------------------*

FORM enq_sbook.
  CALL FUNCTION 'ENQUEUE_ESBOOK'
       EXPORTING
            carrid         = sdyn_book-carrid
            connid         = sdyn_book-connid
            fldate         = sdyn_book-fldate
            bookid         = sdyn_book-bookid
       EXCEPTIONS
            foreign_lock   = 1
            system_failure = 2
            OTHERS         = 3.
  CASE sy-subrc.
    WHEN 0.
    WHEN 1.
      MESSAGE e061.
    WHEN OTHERS.
      MESSAGE e063 WITH sy-subrc.
  ENDCASE.
ENDFORM.                               " ENQ_SBOOK
*&-----------------------------------------------------------------*

*&      Form  DEQ_ALL
*&-----------------------------------------------------------------*

FORM deq_all.
  CALL FUNCTION 'DEQUEUE_ALL'.
ENDFORM.                               " DEQ_ALL
16 F03
*------------------------------------------------------------------*

*   INCLUDE BC414S_BOOKINGSF03
*------------------------------------------------------------------*

*&-----------------------------------------------------------------*

*&      Form  READ_SFLIGHT
*&-----------------------------------------------------------------*

*      -->P_WA_SFLIGHT  text
*      -->P_SYSUBRC     text
*------------------------------------------------------------------*

FORM read_sflight USING p_wa_sflight TYPE sflight
                        p_sysubrc LIKE sy-subrc.
  SELECT SINGLE * FROM sflight INTO p_wa_sflight
         WHERE carrid = sdyn_conn-carrid
         AND   connid = sdyn_conn-connid
         AND   fldate = sdyn_conn-fldate.
  p_sysubrc = sy-subrc.
ENDFORM.                               " READ_SFLIGHT
*&-----------------------------------------------------------------*

*&      Form  READ_SBOOK
*&-----------------------------------------------------------------*

*      -->P_ITAB_BOOK  text
*      -->P_ITAB_CD    text
*------------------------------------------------------------------*

FORM read_sbook USING p_itab_book LIKE itab_book
                      p_itab_cd   LIKE itab_cd.
  TYPES: BEGIN OF wa_custom_type,
           id TYPE scustom-id,
           name TYPE scustom-name,
         END OF wa_custom_type.
  DATA: wa_custom TYPE wa_custom_type,
        itab_custom TYPE STANDARD TABLE OF wa_custom_type
        WITH NON-UNIQUE KEY id,
        wa_book LIKE LINE OF p_itab_book,
        wa_cd   LIKE LINE OF p_itab_cd.
  CLEAR: p_itab_book, p_itab_cd.
* Select customer names in buffer table (array fetch)
  SELECT id name FROM scustom INTO CORRESPONDING FIELDS
         OF TABLE itab_custom.
* Select all bookings on selected flight (array fetch)
  SELECT * FROM sbook INTO CORRESPONDING FIELDS OF TABLE p_itab_book
         WHERE carrid = sdyn_conn-carrid
         AND   connid = sdyn_conn-connid
         AND   fldate = sdyn_conn-fldate.
* read customer names corresponding to customer number from buffer 
* table
  LOOP AT p_itab_book INTO wa_book.
    READ TABLE itab_custom INTO wa_custom WITH TABLE KEY
                                            id = wa_book-customid.
    wa_book-name = wa_custom-name.
    MODIFY p_itab_book FROM wa_book.
    MOVE-CORRESPONDING wa_book TO wa_cd.
    APPEND wa_cd TO p_itab_cd.
  ENDLOOP.
  SORT p_itab_book BY bookid customid.
ENDFORM.                               " READ_SBOOK
*&-----------------------------------------------------------------*

*&      Form  READ_SPFLI
*&-----------------------------------------------------------------*

*      -->P_WA_SPFLI  text
*------------------------------------------------------------------*

FORM read_spfli USING p_wa_spfli TYPE spfli.
  SELECT SINGLE * FROM spfli INTO p_wa_spfli
         WHERE carrid = sdyn_conn-carrid
         AND   connid = sdyn_conn-connid.
  IF sy-subrc <> 0.
    PERFORM deq_all.
    MESSAGE e022 WITH sdyn_conn-carrid sdyn_conn-connid.
  ENDIF.
ENDFORM.                               " READ_SPFLI
17 F04
*------------------------------------------------------------------*

*   INCLUDE BC414S_BOOKINGSF04
*------------------------------------------------------------------*

*&-----------------------------------------------------------------*

*&      Form  SAVE_MODIFIED_BOOKING
*&-----------------------------------------------------------------*

FORM save_modified_booking.
* Modify data on database tables sbook and sflight
  CALL FUNCTION 'UPDATE_SBOOK' IN UPDATE TASK
       EXPORTING
            itab_sbook = itab_sbook_modify.
  PERFORM update_sflight.
ENDFORM.                               " SAVE_MODIFIED_BOOKING
*&-----------------------------------------------------------------*

*&      Form  UPDATE_SFLIGHT
*&-----------------------------------------------------------------*

FORM update_sflight.
  CALL FUNCTION 'UPDATE_SFLIGHT' IN UPDATE TASK
       EXPORTING
            carrier    = wa_sflight-carrid
            connection = wa_sflight-connid
            date       = wa_sflight-fldate.
ENDFORM.                               " UPDATE_SFLIGHT
*&-----------------------------------------------------------------*

*&      Form  SAVE_NEW_BOOKING
*&-----------------------------------------------------------------*

FORM save_new_booking.
* transform amount from foreign currency to local currency (of carrier)
  PERFORM convert_to_loc_currency USING wa_sbook.
* number ranges: Get next number (internal)
  PERFORM number_get_next USING wa_sbook.
* lock booking to be created
  PERFORM enq_sbook.
  CALL FUNCTION 'INSERT_SBOOK' IN UPDATE TASK
       EXPORTING
            wa_sbook = wa_sbook.
  PERFORM update_sflight.
ENDFORM.                               " SAVE_NEW_BOOKING
18 F05
*------------------------------------------------------------------*

*   INCLUDE BC414S_BOOKINGSF05
*------------------------------------------------------------------*

*&-----------------------------------------------------------------*

*&      Form  CONVERT_TO_LOC_CURRENCY
*&-----------------------------------------------------------------*

*      -->P_WA_SBOOK  text
*------------------------------------------------------------------*

FORM convert_to_loc_currency USING p_wa_sbook TYPE sbook.
  SELECT SINGLE currcode FROM scarr INTO p_wa_sbook-loccurkey
         WHERE carrid = p_wa_sbook-carrid.
  CALL FUNCTION 'CONVERT_TO_LOCAL_CURRENCY_N'
       EXPORTING
            client           = sy-mandt
            date             = sy-datum
            foreign_amount   = p_wa_sbook-forcuram
            foreign_currency = p_wa_sbook-forcurkey
            local_currency   = p_wa_sbook-loccurkey
       IMPORTING
            local_amount     = p_wa_sbook-loccuram
       EXCEPTIONS
            no_rate_found    = 1
            overflow         = 2
            no_factors_found = 3
            no_spread_found  = 4
            derived_2_times  = 5
            OTHERS           = 6.
  IF sy-subrc <> 0.
    MESSAGE e080 WITH sy-subrc.
  ENDIF.
ENDFORM.                               " CONVERT_TO_LOC_CURRENCY
19 F06
*------------------------------------------------------------------*

*   INCLUDE BC414S_BOOKINGSF06
*------------------------------------------------------------------*

*&-----------------------------------------------------------------*

*&      Form  CREATE_CHANGE_DOCUMENTS
*&-----------------------------------------------------------------*

FORM create_change_documents.
  LOOP AT itab_sbook_modify INTO sbook.
* read unchanged data from buffer table into *-work area
    READ TABLE itab_cd FROM sbook INTO *sbook.
* define objectid from key fields of sbook
    CONCATENATE sbook-mandt sbook-carrid sbook-connid
                sbook-fldate sbook-bookid sbook-customid
                INTO objectid SEPARATED BY space.
* fill interface parameters of function, which itself is encapsulated
* in form CD_CALL_BC_BOOK
    MOVE: sy-tcode        TO tcode,
          sy-uzeit        TO utime,
          sy-datum        TO udate,
          sy-uname        TO username,
          'U'             TO upd_sbook.
* perform calls the neccessary function to create change document
* 'in update task'
    PERFORM cd_call_bc_book.
  ENDLOOP.
ENDFORM.                               " CREATE_CHANGE_DOCUMENTS

